
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc78894323]Corrections required
0.1 [bookmark: _Toc78894324]Correction to function f_TC_7_1_3_4_1_NR_Step10to23
	Function name
	f_TC_7_1_3_4_1_NR_Step10to23

	Reason for change
	In the current TTCN implementation of 7.1.3.4.1
1) Using the type NR_PDCP_PDUList_Type for the single PDU is leading to mismatch while receiving the PDCP PDU List containing multiple PDUs coming in the same slot at the steps 15
2) At step 15, UE loopback is expected on NR Cell 2 as per prose , but TTCN is expecting loop pack PDCP Data PDU on NR Cell 1 

	Summary of change
	1) For the test case 7.1.3.4.1 , it is proposed to use the data type NR_PDCP_PDU_Type for the expected individual PDUs at steps 15
2) At step 15 of the test case 7.1.3.4.1 , TTCN is corrected to receive the Looped back PDCP Data PDU on NR Cell2

	TTCN module
	NR_TC_Common\7_1_3\PDCP_TC_Common_NR.ttcn

	MCC160 Comment
	



Before Change:
	function f_TC_7_1_3_4_1_NR_Step10to23(inout NR_PDCP_SS_State_Type p_NR_PDCP_State, boolean p_WaitforRRCConCmp := false) runs on NR_BASE_PTC
  {
    var octetstring v_IPData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());
    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;
    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);
    var SubFrameTiming_Type v_TimingDL;
    var SubFrameTiming_Type v_TimingUL;
    var SubFrameTiming_Type v_TimingCurrentSlot;
    var SubFrameTiming_Type v_TimingPreviousSlot;
   // var TimingInfo_Type v_TimingInfo;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu1;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu2;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu3;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu4;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu5;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu6;
    var NR_DRB_COMMON_IND v_ReceivedPdu1;
    var NR_DRB_COMMON_IND v_ReceivedPdu2;
    var NR_DRB_COMMON_IND v_ReceivedPdu3;
    var NR_DRB_COMMON_IND v_ReceivedPdu4;
    var NR_DRB_COMMON_IND v_ReceivedPdu5;
    var NR_DRB_COMMON_IND v_ReceivedPdu6;
    var NR_AbsoluteCellPower_Type v_CellPowerAtT2Cell1_FR1 := -82;
    var NR_AbsoluteCellPower_Type v_CellPowerAtT2Cell1_FR2 := -82;
    var NR_AbsoluteCellPower_Type v_CellPowerAtT2Cell2_FR1 := -88;
    var NR_AbsoluteCellPower_Type v_CellPowerAtT2Cell2_FR2 := -91;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT2 :=  {
            cs_NR_CellPower(nr_Cell1, v_CellPowerAtT2Cell1_FR1, v_CellPowerAtT2Cell1_FR2),
            cs_NR_CellPower(nr_Cell2, v_CellPowerAtT2Cell2_FR1, v_CellPowerAtT2Cell2_FR2)
     }; //@sic R5-195929 sic@
    timer t_Wait:= 5.0;
    
    // for ENDC RRCReconfigurationComplete is not received, just one UL Grant is assigned (112 bits)for STATUS PDU
    // For NR5GC, RRCReconfigurationComplete is received, 2 Grants (2x112 bits)are assigned, one for RRCReconfigurationComplete and one for STATUS PDU

    //@siclog "Step 10" siclog@
    // The SS assigns  UL grant during the Random Access procedure on NR Cell 2 to allow the UE to send only PDCP status report and
    // 5 octets PDCP STATUS REPORT + 3 octets RLC header + 2 octets MAC header + 3oct BSR or padding = 13 octets
    
    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(112, v_NR_UplinkBWP); //@sic R5-198906 sic@
    //@sic R5s201484 sic@
    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell2); // Timing of DL
    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell2, v_TimingDL, 20);
    if ( p_WaitforRRCConCmp) {  // if NR5GC start a periodic grant to receive RRCReconfigurationComplete and PDCP STATUS REPORT  @sic R5s210282 sic@
      f_NR_ULGrantConfiguration_StartPeriodicGrant(nr_Cell2, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20);
    } else {  // if ENDC - only one grant for PDCP STATUS REPORT
      f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell2, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 1);
    }
    
    if ( p_WaitforRRCConCmp) {//@sic R5s210032 sic@
        interleave {//@sic R5s201484 sic@
            //Receive RRCReconfigurationComplete
            []SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell2, cr_38508_RRCReconfigurationComplete)){}
    
            //@siclog "Step 11" siclog@
            //Does the UE send PDCP Control PDUs via RLC-AM RB with the following content to the SS:
            //D/C field = 0 (PDCP control PDU) and PDU Type =000, FMC field = 5?
            []DRB.receive (car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell2,
                                                        p_NR_PDCP_State.DRB_Id,
                                                        cr_RlcBearerRouting_NR(nr_Cell2), //@sic R5s190844 sic@
                                                        -,
                                                        {cr_NR_PDCP_STATUS_REPORT(5)})){}

        }
    }
    else{
        //@siclog "Step 11" siclog@
        //Does the UE send PDCP Control PDUs via RLC-AM RB with the following content to the SS:
        //D/C field = 0 (PDCP control PDU) and PDU Type =000, FMC field = 5?
         DRB.receive (car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell2,
                                                        p_NR_PDCP_State.DRB_Id,
                                                        cr_RlcBearerRouting_NR(nr_Cell2), //@sic R5s190844 sic@
                                                        -,
                                                        {cr_NR_PDCP_STATUS_REPORT(5)}));
    }
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 11");
    
    if ( p_WaitforRRCConCmp) {  // if NR5GC  @sic R5s210282 sic@
        f_NR_ULGrantConfiguration_Stop(nr_Cell2);
    }
    
    if(p_WaitforRRCConCmp){ //@sic R5-204363 sic@
        f_NR_SS_SRBs_DRBs_Release(nr_Cell1, cs_TimingInfo_Now, f_NR_GetActiveDRBs(nr_Cell1)); //@sic R5s200690 sic@
        f_NR_SS_SRBs_DRBs_Config(nr_Cell1, cs_TimingInfo_Now, omit);
    };
    
    //@siclog "Step 12" siclog@
    //The SS generates a PDCP status report message and sends it to UE: D/C field = 0 (PDCP control PDU)
    //and PDU Type =000, FMC field = 3
    DRB.send (cas_NR_DRB_COMMON_REQ_PDCP_PDUList (nr_Cell2,
                                               p_NR_PDCP_State.DRB_Id,
                                               -,
                                               {cs_NR_PDCP_STATUS_REPORT(2)})); //@sic R5-196261 sic@
                                               
   //@siclog "Step 13" siclog@
   // The SS starts the UL default grant transmission.
   f_NR_ULGrantConfiguration_Start(nr_Cell2);
   
   //@siclog "Step 15" siclog@
   //@sic R5-213462 sic@
    p_NR_PDCP_State.RX_NEXT:= 2;//@sic R5-196261 sic@
   
    v_ExpectedPdu1:= {f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT)};
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu2:= {f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT)};
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu3:= {f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT)};
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu4:= (v_ExpectedPdu1, v_ExpectedPdu2);
    v_ExpectedPdu5:= (v_ExpectedPdu2, v_ExpectedPdu3);
    v_ExpectedPdu6:= (v_ExpectedPdu1, v_ExpectedPdu2, v_ExpectedPdu3);
    
  alt {
      // receive PDU1
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu1)) -> value v_ReceivedPdu1
        {
          v_TimingPreviousSlot := v_ReceivedPdu1.Common.TimingInfo.SubFrame;
          alt {
            // receive PDU2
            [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu2)) -> value v_ReceivedPdu2
              {
                v_TimingCurrentSlot := v_ReceivedPdu2.Common.TimingInfo.SubFrame;
                f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
                v_TimingPreviousSlot:= v_TimingCurrentSlot;
              }
                // receive PDU3
            [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu3)) -> value v_ReceivedPdu3
                {
                  v_TimingCurrentSlot := v_ReceivedPdu3.Common.TimingInfo.SubFrame;
                  f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
                  f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");
                }
            // receive PDU2 and PDU3
            [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu5)) -> value v_ReceivedPdu5
              {
                v_TimingCurrentSlot := v_ReceivedPdu5.Common.TimingInfo.SubFrame;
                f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
                f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");
              }
           }
         }
      // receive PDU1 and PDU2
      [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu4)) -> value v_ReceivedPdu4
        {
          v_TimingPreviousSlot := v_ReceivedPdu4.Common.TimingInfo.SubFrame;
            // receive PDU3
           DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu3)) -> value v_ReceivedPdu3;
           v_TimingCurrentSlot := v_ReceivedPdu3.Common.TimingInfo.SubFrame;
           f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
           f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");
         }
      // receive PDU1, PDU2 and PDU3
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu6)) -> value v_ReceivedPdu6
        {
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");
        }
    }
    
   //@siclog "Step 16" siclog@
    //The SS sends the PDCP Data PDU#5
    p_NR_PDCP_State.TX_NEXT:= 5;
    f_NR_PDCP_Send(p_NR_PDCP_State, nr_Cell2, v_IPData);
   
    //@siclog "Step 17" siclog@
   //The UE sends the PDCP Data PDU#5
    f_NR_PDCP_Recv(p_NR_PDCP_State, nr_Cell2, v_IPData, cr_RlcBearerRouting_NR(nr_Cell2));
    
    //@siclog "Step 20" siclog@
    //SS transmits a PDCP Data PDU containing one PDCP SDU#7
    p_NR_PDCP_State.TX_NEXT:= 7;
    p_NR_PDCP_State.RX_NEXT:= 7;
    //The SS sends the PDCP Data PDU
    f_NR_PDCP_Send(p_NR_PDCP_State, nr_Cell2, v_IPData);
   
   //@siclog "Step 21" siclog@
   //Check: Does the UE transmit a PDCP DATA PDU#7 on NR Cell 2?
    t_Wait.start;
    alt {
      [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList(nr_Cell2, p_NR_PDCP_State.DRB_Id)) {
          t_Wait.stop;
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 21");
        }
      [] t_Wait.timeout{}
    }
    
    //@siclog "Step 21A" siclog@
    f_NR_ULGrantConfiguration_Stop(nr_Cell1); //@sic R5-190752 sic@
   //@siclog "Step 22" siclog@
   //The SS changes NR Cell 1 and NR Cell 2 parameters according to row "T2" in Table 7.1.3.4.1.3.1-1.
   f_NR_SetCellPowerList(v_CellPowerList_AtT2);
  }



After Change:
	  function f_TC_7_1_3_4_1_NR_Step10to23(inout NR_PDCP_SS_State_Type p_NR_PDCP_State, boolean p_WaitforRRCConCmp := false) runs on NR_BASE_PTC
  {
    var octetstring v_IPData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());
    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;
    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);
    var SubFrameTiming_Type v_TimingDL;
    var SubFrameTiming_Type v_TimingUL;
    var SubFrameTiming_Type v_TimingCurrentSlot;
    var SubFrameTiming_Type v_TimingPreviousSlot;
   // var TimingInfo_Type v_TimingInfo;
    var template (present) NR_PDCP_PDU_Type v_ExpectedPdu1; //WA#WI=861440
    var template (present) NR_PDCP_PDU_Type v_ExpectedPdu2; //WA#WI=861440
    var template (present) NR_PDCP_PDU_Type v_ExpectedPdu3; //WA#WI=861440
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu4;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu5;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu6;
    var NR_DRB_COMMON_IND v_ReceivedPdu1;
    var NR_DRB_COMMON_IND v_ReceivedPdu2;
    var NR_DRB_COMMON_IND v_ReceivedPdu3;
    var NR_DRB_COMMON_IND v_ReceivedPdu4;
    var NR_DRB_COMMON_IND v_ReceivedPdu5;
    var NR_DRB_COMMON_IND v_ReceivedPdu6;
    var NR_AbsoluteCellPower_Type v_CellPowerAtT2Cell1_FR1 := -82;
    var NR_AbsoluteCellPower_Type v_CellPowerAtT2Cell1_FR2 := -82;
    var NR_AbsoluteCellPower_Type v_CellPowerAtT2Cell2_FR1 := -88;
    var NR_AbsoluteCellPower_Type v_CellPowerAtT2Cell2_FR2 := -91;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT2 :=  {
            cs_NR_CellPower(nr_Cell1, v_CellPowerAtT2Cell1_FR1, v_CellPowerAtT2Cell1_FR2),
            cs_NR_CellPower(nr_Cell2, v_CellPowerAtT2Cell2_FR1, v_CellPowerAtT2Cell2_FR2)
     }; //@sic R5-195929 sic@
    timer t_Wait:= 5.0;
    
    // for ENDC RRCReconfigurationComplete is not received, just one UL Grant is assigned (112 bits)for STATUS PDU
    // For NR5GC, RRCReconfigurationComplete is received, 2 Grants (2x112 bits)are assigned, one for RRCReconfigurationComplete and one for STATUS PDU

    //@siclog "Step 10" siclog@
    // The SS assigns  UL grant during the Random Access procedure on NR Cell 2 to allow the UE to send only PDCP status report and
    // 5 octets PDCP STATUS REPORT + 3 octets RLC header + 2 octets MAC header + 3oct BSR or padding = 13 octets
    
    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(112, v_NR_UplinkBWP); //@sic R5-198906 sic@
    //@sic R5s201484 sic@
    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell2); // Timing of DL
    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell2, v_TimingDL, 20);
    if ( p_WaitforRRCConCmp) {  // if NR5GC start a periodic grant to receive RRCReconfigurationComplete and PDCP STATUS REPORT  @sic R5s210282 sic@
      f_NR_ULGrantConfiguration_StartPeriodicGrant(nr_Cell2, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20);
    } else {  // if ENDC - only one grant for PDCP STATUS REPORT
      f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell2, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 1);
    }
    
    if ( p_WaitforRRCConCmp) {//@sic R5s210032 sic@
        interleave {//@sic R5s201484 sic@
            //Receive RRCReconfigurationComplete
            []SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell2, cr_38508_RRCReconfigurationComplete)){}
    
            //@siclog "Step 11" siclog@
            //Does the UE send PDCP Control PDUs via RLC-AM RB with the following content to the SS:
            //D/C field = 0 (PDCP control PDU) and PDU Type =000, FMC field = 5?
            []DRB.receive (car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell2,
                                                        p_NR_PDCP_State.DRB_Id,
                                                        cr_RlcBearerRouting_NR(nr_Cell2), //@sic R5s190844 sic@
                                                        -,
                                                        {cr_NR_PDCP_STATUS_REPORT(5)})){}

        }
    }
    else{
        //@siclog "Step 11" siclog@
        //Does the UE send PDCP Control PDUs via RLC-AM RB with the following content to the SS:
        //D/C field = 0 (PDCP control PDU) and PDU Type =000, FMC field = 5?
         DRB.receive (car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell2,
                                                        p_NR_PDCP_State.DRB_Id,
                                                        cr_RlcBearerRouting_NR(nr_Cell2), //@sic R5s190844 sic@
                                                        -,
                                                        {cr_NR_PDCP_STATUS_REPORT(5)}));
    }
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 11");
    
    if ( p_WaitforRRCConCmp) {  // if NR5GC  @sic R5s210282 sic@
        f_NR_ULGrantConfiguration_Stop(nr_Cell2);
    }
    
    if(p_WaitforRRCConCmp){ //@sic R5-204363 sic@
        f_NR_SS_SRBs_DRBs_Release(nr_Cell1, cs_TimingInfo_Now, f_NR_GetActiveDRBs(nr_Cell1)); //@sic R5s200690 sic@
        f_NR_SS_SRBs_DRBs_Config(nr_Cell1, cs_TimingInfo_Now, omit);
    };
    
    //@siclog "Step 12" siclog@
    //The SS generates a PDCP status report message and sends it to UE: D/C field = 0 (PDCP control PDU)
    //and PDU Type =000, FMC field = 3
    DRB.send (cas_NR_DRB_COMMON_REQ_PDCP_PDUList (nr_Cell2,
                                               p_NR_PDCP_State.DRB_Id,
                                               -,
                                               {cs_NR_PDCP_STATUS_REPORT(2)})); //@sic R5-196261 sic@
                                               
   //@siclog "Step 13" siclog@
   // The SS starts the UL default grant transmission.
   f_NR_ULGrantConfiguration_Start(nr_Cell2);
   
   //@siclog "Step 15" siclog@
   //@sic R5-213462 sic@
    p_NR_PDCP_State.RX_NEXT:= 2;//@sic R5-196261 sic@
   
    v_ExpectedPdu1:= f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT); //WA#WI=861440
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu2:= f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT); //WA#WI=861440
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu3:= f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT); //WA#WI=861440
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu4:= {v_ExpectedPdu1, v_ExpectedPdu2}; //WA#WI=861440
    v_ExpectedPdu5:= {v_ExpectedPdu2, v_ExpectedPdu3}; //WA#WI=861440
    v_ExpectedPdu6:= {v_ExpectedPdu1, v_ExpectedPdu2, v_ExpectedPdu3}; //WA#WI=861440
    
  alt {
      // receive PDU1
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell2, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell2), -, {v_ExpectedPdu1})) -> value v_ReceivedPdu1 //WA#WI=861440
        {
          v_TimingPreviousSlot := v_ReceivedPdu1.Common.TimingInfo.SubFrame;
          alt {
            // receive PDU2
            [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell2, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell2), -, {v_ExpectedPdu2})) -> value v_ReceivedPdu2 //WA#WI=861440
              {
                v_TimingCurrentSlot := v_ReceivedPdu2.Common.TimingInfo.SubFrame;
                f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
                v_TimingPreviousSlot:= v_TimingCurrentSlot;
              }
                // receive PDU3
            [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell2, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell2), -, {v_ExpectedPdu3})) -> value v_ReceivedPdu3 //WA#WI=861440
                {
                  v_TimingCurrentSlot := v_ReceivedPdu3.Common.TimingInfo.SubFrame;
                  f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
                  f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");
                }
            // receive PDU2 and PDU3
            [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell2, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell2), -, v_ExpectedPdu5)) -> value v_ReceivedPdu5 //WA#WI=861440
              {
                v_TimingCurrentSlot := v_ReceivedPdu5.Common.TimingInfo.SubFrame;
                f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
                f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");
              }
           }
         }
      // receive PDU1 and PDU2
      [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell2, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell2), -, v_ExpectedPdu4)) -> value v_ReceivedPdu4 //WA#WI=861440
        {
          v_TimingPreviousSlot := v_ReceivedPdu4.Common.TimingInfo.SubFrame;
            // receive PDU3
           DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell2, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell2), -, {v_ExpectedPdu3})) -> value v_ReceivedPdu3; //WA#WI=861440
           v_TimingCurrentSlot := v_ReceivedPdu3.Common.TimingInfo.SubFrame;
           f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
           f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");
         }
      // receive PDU1, PDU2 and PDU3
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell2, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell2), -, v_ExpectedPdu6)) -> value v_ReceivedPdu6 //WA#WI=861440
        {
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");
        }
    }
    
   //@siclog "Step 16" siclog@
    //The SS sends the PDCP Data PDU#5
    p_NR_PDCP_State.TX_NEXT:= 5;
    f_NR_PDCP_Send(p_NR_PDCP_State, nr_Cell2, v_IPData);
   
    //@siclog "Step 17" siclog@
   //The UE sends the PDCP Data PDU#5
    f_NR_PDCP_Recv(p_NR_PDCP_State, nr_Cell2, v_IPData, cr_RlcBearerRouting_NR(nr_Cell2));
    
    //@siclog "Step 20" siclog@
    //SS transmits a PDCP Data PDU containing one PDCP SDU#7
    p_NR_PDCP_State.TX_NEXT:= 7;
    p_NR_PDCP_State.RX_NEXT:= 7;
    //The SS sends the PDCP Data PDU
    f_NR_PDCP_Send(p_NR_PDCP_State, nr_Cell2, v_IPData);
   
   //@siclog "Step 21" siclog@
   //Check: Does the UE transmit a PDCP DATA PDU#7 on NR Cell 2?
    t_Wait.start;
    alt {
      [] DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList(nr_Cell2, p_NR_PDCP_State.DRB_Id)) {
          t_Wait.stop;
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 21");
        }
      [] t_Wait.timeout{}
    }
    
    //@siclog "Step 21A" siclog@
    f_NR_ULGrantConfiguration_Stop(nr_Cell1); //@sic R5-190752 sic@
   //@siclog "Step 22" siclog@
   //The SS changes NR Cell 1 and NR Cell 2 parameters according to row "T2" in Table 7.1.3.4.1.3.1-1.
   f_NR_SetCellPowerList(v_CellPowerList_AtT2);
  }


0.2 [bookmark: _Toc78894325]Correction to function f_TC_7_1_3_4_1_NR_Step25to30
	Function name
	f_TC_7_1_3_4_1_NR_Step25to30

	Reason for change
	In the current TTCN implementation
Using the type NR_PDCP_PDUList_Type for the single PDU is leading to mismatch while receiving the PDCP PDU List containing multiple PDUs coming in the same slot at the steps 30 and 31


	Summary of change
	For the test case 7.1.3.4.1 , it is proposed to use the data type NR_PDCP_PDU_Type for the expected individual PDUs at steps 30 and 31


	TTCN module
	\ttcn\develop\NR_TC_Common\7_1_3\PDCP_TC_Common_NR.ttcn

	MCC160 Comment
	



Before Change:
	function f_TC_7_1_3_4_1_NR_Step25to30(inout NR_PDCP_SS_State_Type p_NR_PDCP_State,  boolean p_WaitforRRCConCmp := false) runs on NR_BASE_PTC
  {
    var octetstring v_IPData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());
    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;
    var SubFrameTiming_Type v_TimingDL;
    var SubFrameTiming_Type v_TimingUL;
    var SubFrameTiming_Type v_TimingCurrentSlot;
    var SubFrameTiming_Type v_TimingPreviousSlot;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu1;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu2;
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu3;
    var NR_DRB_COMMON_IND v_ReceivedPdu1;
    var NR_DRB_COMMON_IND v_ReceivedPdu2;
    var NR_DRB_COMMON_IND v_ReceivedPdu3;
    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);
    timer t_Wait:= 5.0;
   
    // for ENDC RRCReconfigurationComplete is not received, jut one UL Grant is assigned (112 bits)for STATUS PDU
    // For NR5GC, RRCReconfigurationComplete is received, 2 Grants (2x112 bits)are assigned, one for RRCReconfigurationComplete and one for STATUS PDU
 
     //@siclog "Step 24" siclog@
    // The SS assigns UL grant during the Random Access procedure on NR Cell 1 to allow the UE to send only PDCP status report.
    // 5 octets PDCP STATUS REPORT + 3 octets RLC header + 2 octets MAC header = 10 octets
    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(112, v_NR_UplinkBWP); //@sic R5-198906 sic@
    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // Timing of DL
    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL, 20);
     if ( p_WaitforRRCConCmp) {  // if NR5GC start a periodic grant to receive RRCReconfigurationComplete and PDCP STATUS REPORT @sic R5s210282 sic@
      f_NR_ULGrantConfiguration_StartPeriodicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20);
    } else {  // if ENDC
      f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 1);
    }

    if ( p_WaitforRRCConCmp) {//@sic R5s210032 sic@
        //EXCEPTION: Steps 25 and 26 can occur in any order. (Note 7)
        interleave {//@sic R5s201484 sic@
            //@siclog "Step 25" siclog@
            //The UE transmits a NR RRCReconfigurationComplete message.(Note 4)
            []SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete)){}
    
            //@siclog "Step 26" siclog@
            //Does the UE send PDCP Control PDUs via RLC-AM RB with the following content to the SS:
            //D/C field = 0 (PDCP control PDU) and PDU Type =000, FMC field = 5?
            []DRB.receive (car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1,
                                                    p_NR_PDCP_State.DRB_Id,
                                                    cr_RlcBearerRouting_NR(nr_Cell1), //@sic R5s190844 sic@
                                                    -,
                                                    {cr_NR_PDCP_STATUS_REPORT(6,'80'O)})){}

        }
     
     }
     
    else{
        //@siclog "Step 26" siclog@
        //Does the UE send PDCP Control PDUs via RLC-AM RB with the following content to the SS:
        //D/C field = 0 (PDCP control PDU) and PDU Type =000, FMC field = 5?
        DRB.receive (car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1,
                                                    p_NR_PDCP_State.DRB_Id,
                                                    cr_RlcBearerRouting_NR(nr_Cell1), //@sic R5s190844 sic@
                                                    -,
                                                    {cr_NR_PDCP_STATUS_REPORT(6,'80'O)}));
    }
     
    if(p_WaitforRRCConCmp){ //@sic R5-204363 sic@
        f_NR_ULGrantConfiguration_Stop(nr_Cell1);    // @sic R5s210282 sic@
        f_NR_SS_SRBs_DRBs_Release(nr_Cell2, cs_TimingInfo_Now, f_NR_GetActiveDRBs(nr_Cell2)); //@sic R5s200690 sic@
        f_NR_SS_SRBs_DRBs_Config(nr_Cell2, cs_TimingInfo_Now, omit);
    };
   //@siclog "Step 27" siclog@
   //The SS generates a PDCP status report message and sends it to UE: D/C field = 0 (PDCP control PDU) and PDU Type =000, FMC field = 6 on NR Cell 1.
    DRB.send (cas_NR_DRB_COMMON_REQ_PDCP_PDUList (nr_Cell1,
                                               p_NR_PDCP_State.DRB_Id,
                                               -,
                                               {cs_NR_PDCP_STATUS_REPORT(6)}));
                                               

    //@siclog "Step 28" siclog@
    // The SS starts the UL default grant transmission.
    f_NR_ULGrantConfiguration_Start(nr_Cell1);
    
    //@siclog "Step 28A" siclog@
    //The SS sends the PDCP Data PDU#5
    p_NR_PDCP_State.TX_NEXT:= 5;
    f_NR_PDCP_Send(p_NR_PDCP_State, nr_Cell1, v_IPData); //@sic R5-190752 sic@

    //@siclog "Step 28B" siclog@
    //Check: Does the UE transmit a PDCP Data PDU via RLC-AM RB with the following content back to the SS?
    t_Wait.start; //@sic R5-190752 sic@
    alt {
      [] DRB.receive((car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1,
                                                         p_NR_PDCP_State.DRB_Id,
                                                         cr_RlcBearerRouting_NR(nr_Cell1)))) {
          t_Wait.stop;
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 28B");
        }
      [] t_Wait.timeout{}
    }
    
    //@siclog "Step 29" siclog@
    //SS transmits a PDCP Data PDU containing one PDCP SDU#6
    p_NR_PDCP_State.TX_NEXT:= 6;
    p_NR_PDCP_State.RX_NEXT:= 6;
    f_NR_PDCP_Send(p_NR_PDCP_State, nr_Cell1, v_IPData);
   
   //@siclog "Steps 30 - 31" siclog@
   //Check: Does the UE transmit a PDCP Data PDU via RLC-AM RB with the following content back to the SS? D/C field = 1 (PDCP Data PDU) and PDCP SN=6 on NR Cell 1.
   //Note: Data is previously received packet in PDCP Data PDU#6.(Note 9)
   //Check: Does the UE transmit PDCP Data PDU via RLC-AM RB with the following content back to the SS? D/C field = 1 (PDCP Data PDU) and PDCP SN=7 on NR Cell 1.
   //Note: Data is previously received packet in PDCP Data PDU#7(Note 9).
    //@sic R5-213462  sic@
    v_ExpectedPdu1:= {f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT)};
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu2:= {f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT)};
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu3:= (v_ExpectedPdu1,v_ExpectedPdu2);
    
    alt {
      // receive PDU1
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu1)) -> value v_ReceivedPdu1
        {
          v_TimingPreviousSlot := v_ReceivedPdu1.Common.TimingInfo.SubFrame;
          // receive PDU2
          DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu2)) -> value v_ReceivedPdu2;
          v_TimingCurrentSlot := v_ReceivedPdu2.Common.TimingInfo.SubFrame;
          f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 30-31");
        }
      // receive PDU1 and PDU2
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu3))-> value v_ReceivedPdu3
        {
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 30-31");
        }
    }
  }



After Change:
	function f_TC_7_1_3_4_1_NR_Step25to30(inout NR_PDCP_SS_State_Type p_NR_PDCP_State,  boolean p_WaitforRRCConCmp := false) runs on NR_BASE_PTC
  {
    var octetstring v_IPData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());
    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;
    var SubFrameTiming_Type v_TimingDL;
    var SubFrameTiming_Type v_TimingUL;
    var SubFrameTiming_Type v_TimingCurrentSlot;
    var SubFrameTiming_Type v_TimingPreviousSlot;
    var template (present) NR_PDCP_PDU_Type  v_ExpectedPdu1; //WA#WI=861440
    var template (present) NR_PDCP_PDU_Type  v_ExpectedPdu2; //WA#WI=861440
    var template (present) NR_PDCP_PDUList_Type v_ExpectedPdu3;
    var NR_DRB_COMMON_IND v_ReceivedPdu1;
    var NR_DRB_COMMON_IND v_ReceivedPdu2;
    var NR_DRB_COMMON_IND v_ReceivedPdu3;
    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);
    timer t_Wait:= 5.0;
   
    // for ENDC RRCReconfigurationComplete is not received, jut one UL Grant is assigned (112 bits)for STATUS PDU
    // For NR5GC, RRCReconfigurationComplete is received, 2 Grants (2x112 bits)are assigned, one for RRCReconfigurationComplete and one for STATUS PDU
 
     //@siclog "Step 24" siclog@
    // The SS assigns UL grant during the Random Access procedure on NR Cell 1 to allow the UE to send only PDCP status report.
    // 5 octets PDCP STATUS REPORT + 3 octets RLC header + 2 octets MAC header = 10 octets
    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(112, v_NR_UplinkBWP); //@sic R5-198906 sic@
    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1); // Timing of DL
    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL, 20);
     if ( p_WaitforRRCConCmp) {  // if NR5GC start a periodic grant to receive RRCReconfigurationComplete and PDCP STATUS REPORT @sic R5s210282 sic@
      f_NR_ULGrantConfiguration_StartPeriodicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20);
    } else {  // if ENDC
      f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 20, 1);
    }

    if ( p_WaitforRRCConCmp) {//@sic R5s210032 sic@
        //EXCEPTION: Steps 25 and 26 can occur in any order. (Note 7)
        interleave {//@sic R5s201484 sic@
            //@siclog "Step 25" siclog@
            //The UE transmits a NR RRCReconfigurationComplete message.(Note 4)
            []SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete)){}
    
            //@siclog "Step 26" siclog@
            //Does the UE send PDCP Control PDUs via RLC-AM RB with the following content to the SS:
            //D/C field = 0 (PDCP control PDU) and PDU Type =000, FMC field = 5?
            []DRB.receive (car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1,
                                                    p_NR_PDCP_State.DRB_Id,
                                                    cr_RlcBearerRouting_NR(nr_Cell1), //@sic R5s190844 sic@
                                                    -,
                                                    {cr_NR_PDCP_STATUS_REPORT(6,'80'O)})){}

        }
     
     }
     
    else{
        //@siclog "Step 26" siclog@
        //Does the UE send PDCP Control PDUs via RLC-AM RB with the following content to the SS:
        //D/C field = 0 (PDCP control PDU) and PDU Type =000, FMC field = 5?
        DRB.receive (car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1,
                                                    p_NR_PDCP_State.DRB_Id,
                                                    cr_RlcBearerRouting_NR(nr_Cell1), //@sic R5s190844 sic@
                                                    -,
                                                    {cr_NR_PDCP_STATUS_REPORT(6,'80'O)}));
    }
     
    if(p_WaitforRRCConCmp){ //@sic R5-204363 sic@
        f_NR_ULGrantConfiguration_Stop(nr_Cell1);    // @sic R5s210282 sic@
        f_NR_SS_SRBs_DRBs_Release(nr_Cell2, cs_TimingInfo_Now, f_NR_GetActiveDRBs(nr_Cell2)); //@sic R5s200690 sic@
        f_NR_SS_SRBs_DRBs_Config(nr_Cell2, cs_TimingInfo_Now, omit);
    };
   //@siclog "Step 27" siclog@
   //The SS generates a PDCP status report message and sends it to UE: D/C field = 0 (PDCP control PDU) and PDU Type =000, FMC field = 6 on NR Cell 1.
    DRB.send (cas_NR_DRB_COMMON_REQ_PDCP_PDUList (nr_Cell1,
                                               p_NR_PDCP_State.DRB_Id,
                                               -,
                                               {cs_NR_PDCP_STATUS_REPORT(6)}));
                                               

    //@siclog "Step 28" siclog@
    // The SS starts the UL default grant transmission.
    f_NR_ULGrantConfiguration_Start(nr_Cell1);
    
    //@siclog "Step 28A" siclog@
    //The SS sends the PDCP Data PDU#5
    p_NR_PDCP_State.TX_NEXT:= 5;
    f_NR_PDCP_Send(p_NR_PDCP_State, nr_Cell1, v_IPData); //@sic R5-190752 sic@

    //@siclog "Step 28B" siclog@
    //Check: Does the UE transmit a PDCP Data PDU via RLC-AM RB with the following content back to the SS?
    t_Wait.start; //@sic R5-190752 sic@
    alt {
      [] DRB.receive((car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1,
                                                         p_NR_PDCP_State.DRB_Id,
                                                         cr_RlcBearerRouting_NR(nr_Cell1)))) {
          t_Wait.stop;
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 28B");
        }
      [] t_Wait.timeout{}
    }
    
    //@siclog "Step 29" siclog@
    //SS transmits a PDCP Data PDU containing one PDCP SDU#6
    p_NR_PDCP_State.TX_NEXT:= 6;
    p_NR_PDCP_State.RX_NEXT:= 6;
    f_NR_PDCP_Send(p_NR_PDCP_State, nr_Cell1, v_IPData);
   
   //@siclog "Steps 30 - 31" siclog@
   //Check: Does the UE transmit a PDCP Data PDU via RLC-AM RB with the following content back to the SS? D/C field = 1 (PDCP Data PDU) and PDCP SN=6 on NR Cell 1.
   //Note: Data is previously received packet in PDCP Data PDU#6.(Note 9)
   //Check: Does the UE transmit PDCP Data PDU via RLC-AM RB with the following content back to the SS? D/C field = 1 (PDCP Data PDU) and PDCP SN=7 on NR Cell 1.
   //Note: Data is previously received packet in PDCP Data PDU#7(Note 9).
    //@sic R5-213462  sic@
    v_ExpectedPdu1:= f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT); //WA#WI=861440
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu2:= f_NR_PDCP_PDU_Get(p_NR_PDCP_State.PDCP_SN_Size, p_NR_PDCP_State.RX_NEXT); //WA#WI=861440
    f_NR_PDCP_Increment_NEXT_RX(p_NR_PDCP_State,1);
    v_ExpectedPdu3:= {v_ExpectedPdu1,v_ExpectedPdu2}; //WA#WI=861440
    
    alt {
      // receive PDU1
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, {v_ExpectedPdu1})) -> value v_ReceivedPdu1 //WA#WI=861440
        {
          v_TimingPreviousSlot := v_ReceivedPdu1.Common.TimingInfo.SubFrame;
          // receive PDU2
          DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, {v_ExpectedPdu2})) -> value v_ReceivedPdu2; //WA#WI=861440
          v_TimingCurrentSlot := v_ReceivedPdu2.Common.TimingInfo.SubFrame;
          f_NR_SubFrameTiming_DifferentSlots (v_TimingCurrentSlot, v_TimingPreviousSlot);
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 30-31");
        }
      // receive PDU1 and PDU2
      []  DRB.receive(car_NR_DRB_COMMON_IND_PDCP_PDUList_DC(nr_Cell1, p_NR_PDCP_State.DRB_Id, cr_RlcBearerRouting_NR(nr_Cell1), -, v_ExpectedPdu3))-> value v_ReceivedPdu3
        {
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Steps 30-31");
        }
    }
  }    



